Schneller und komfortabler Ersatz für 'DLookup()'
Versionen: Access 2007, 2003, 2002/XP und 2000

Die Funktion "DLookup()" hat verschiedene Nachteile: Sie ist recht langsam, liefert nur den ersten, zum Suchkriterium passenden Datensatz ohne Berücksichtigung einer Sortierung und ergibt eventuell falsche Werte, wenn das zu durchsuchende Feld in einem Datensatz den Inhalt "Null" hat. Darüber hinaus kann es bei intensivem Einsatz von "DLookup()" zur missverständlichen Fehlermeldung "Es können keine weiteren Datenbanken geöffnet werden" kommen: DLookup gibt Tabellenhandles mitunter nicht korrekt frei, so dass der Pool verfügbarer Handles früher oder später erschöpft ist. In diesem Tipp der Woche stellen wir eine schnelle und sichere Alternative vor.

Die oben erwähnten Nachteile von "DLookup()" lassen sich durch den Einsatz einer benutzerdefinierten Funktion kompensieren, die wir in Anlehnung an API-Funktionen mit erweitertem Funktionsumfang "DLookupEx()" genannt haben. Diese Funktion arbeitet schneller und zuverlässiger, als "DLookup()", da Sie per "ForwardOnly"-Recordset mit geringstmöglichem Overhead auf die Tabelle zugreift und alle Objekte schließt sowie dereferenziert. Weiterhin haben Sie die Möglichkeit, als zusätzlichen vierten Parameter ein Feld für die Sortierung zu spezifizieren, so dass eine definierte Reihenfolge der zu durchsuchenden Datensätze gewährleistet ist.

Um die Funktion in Ihre Datenbank zu integrieren, gehen Sie folgendermaßen vor:

1. Wechseln Sie mit Alt+F11 in die VBA-Entwicklungsumgebung und legen ein neues Modul an, das Sie "modDLookupEx" benennen.

2. Wählen Sie das Menü Extras-Verweise an und setzen Sie eine Referenz auf "Microsoft DAO 3.6 Object Library".

3. Kopieren Sie die folgende Funktion in das Modul:

Function DLookupEx(Expr As String, _
                Domain As String, _
                Optional Criteria, _
                Optional OrderBy)
  Dim db As DAO.Database
  Dim rs As DAO.Recordset
  Dim strSQL As String

  On Error GoTo DLEx_Error
  strSQL = "SELECT TOP 1 " & Expr & " FROM " & Domain
  If Not IsMissing(Criteria) Then
      strSQL = strSQL & " WHERE " & Criteria
  End If
  If Not IsMissing(OrderBy) Then
      strSQL = strSQL & " ORDER BY " & OrderBy
  End If
  strSQL = strSQL & ";"

  Set db = CurrentDb()
  Set rs = db.OpenRecordset(strSQL, dbOpenForwardOnly)
  If rs.RecordCount = 0 Then
    DLookupEx = Null
  Else
    DLookupEx = rs(0)
  End If
  rs.Close

DLEx_Exit:
  On Error Resume Next
  rs.Close
  Set rs = Nothing
  Set db = Nothing
  Exit Function

DLEx_Error:
  MsgBox "Bei der Ausführung von 'DLookupEx' ist " & _
         "folgender Fehler aufgetreten: " & _
         Err.Description & " (" & Err.Number & ")", _
         vbOKOnly + vbExclamation, "!!! Problem !!!"
  Resume DLEx_Exit

End Function

Um die Funktion einmal zu testen, drücken Sie Strg+G für das Direktfenster und geben Sie dort eine Anweisung wie die folgende ein:

? DLookupEx("Firma", _
            "Kunden", _
            "Land= 'Deutschland'", _
            "Land, Firma")

Diese Anweisung gibt den Inhalt des Feldes "Firma" aus dem ersten Datensatz der Tabelle "Kunden" zurück, bei dem der Inhalt des Feldes "Land" dem Begriff "Deutschland" entspricht. Die Datensätze werden vor der Suche zunächst nach Land und innerhalb des Landes nach Firma sortiert.

Die Funktionsweise von "DLookupEx()" ist schnell erklärt: Im ersten Bereich wird eine SQL-Anweisung mit den Basisanweisungen "Select Top 1" und "From" zusammengesetzt und je nach Vorhandensein von Parametern um "Where"- und "Order by"-Klauseln erweitert. "Top 1" sorgt dafür, dass das Ergebnis analog zum originalen "DLookup()" nur einen Datensatz umfasst. Der zweite Bereich öffnet dann den Recordset und setzt das Funktionsergebnis entweder auf "Null", wenn kein passender Datensatz vorliegt oder auf den Inhalt des ersten Feldes des Recordsets. Zum Abschluss werden der Recordset geschlossen und die Objektvariablen dereferenziert.
